Analytic Methods for Combining HINTS Puerto Rico and HINTS 2007

This text provides instruction on the correct way to combine HINTS Puerto Rico with HINTS 2007 and also furnishes the proper Stata syntax to analyze the combined data set. The method outlined here may also be used to combine other replicate weight dataset that have differing amounts of replicate weights. For more detailed information on combining datasets, see http://hints.cancer.gov/docs/HINTS_Data_Users_Handbook-2008.pdf. 

We begin with some general recommendations on combining survey datasets. All variables that will be used in the analysis should first be examined within each separate dataset, using the formats that are specific to that year. This is important because it is not uncommon for subtle changes to be enacted between survey year iterations. The following should always be checked when data is first combined:
1. Variable names, values, and formatting

2. Variable universes, skip patterns, and labels
In some cases, these will be uniform within different survey iterations and no changes will be needed. In other cases, variable recodes will be necessary to make the definitions consistent across years. A good way to verify consistency is to check descriptive statistics (frequencies, means, etc.) for the variables or recodes for drastic differences. 


Once the variables and recodes are constructed, we must modify the base and replicate weights. Unless weighted sample counts (i.e., population estimates) are desired, no changes are required to be made to the base weights. For instance, when combining two national surveys of similar sample size, a good approximation would be to divide the weights by two. We will not make any adjustments in our case, it will suffice to say that our population is the United States and Puerto Rico. For a general method on the proper reweighting of pooled surveys, see Korn and Graubard (1999).

We are now ready to define a new set of replicate weights within our pooled data set. Since HINTS 2007 has 50 replicate weights and HINTS PR contains 48 replicate weights, the resulting pooled data set should contain 98 replicate weights. For observations from HINTS 2007, the first 50 replicate weights will be the same as before (cwgt1-cwgt50), while replicate weights 51-98 will be equal to the base weight (cwgt0). On the other hand, for observations from Puerto Rico, the first 50 replicate weights will be the base weight (r12wgt0), while replicate weights 51-98 will be equivalent to the 48 original replicate weights (r12wgt1-r12wgt48). The example code below uses the full HINTS 2007 dataset (mail and phone modes combined). Since the numbers of jackknife replicate weights differ, each dataset is analyzed with a different jackknife multiplier: 0.98 in HINTS 2007 and 0.83333 in HINTS Puerto Rico. Therefore, additional syntax is required to properly apply the correct jackknife multiplier to each replicate weight.
*Sample code to combine and analyze the 2007 hints mainland data with the Puerto Rico data

*Step 1: Combine two data files
set memory 512m

* Read in HINTS 2007 data

use "<insert file path name>\hints2007.dta", clear

* Create the analysis variable

recode hc15useinternet (1=1) (2=0), generate(useinternet_r)

* Create the full sampling weight for combined data

generate double nfwgt0 = cwgt0

* Create the replicate weights for combined data

foreach i of numlist 1/50 {


generate double nfwgt`i' = cwgt`i'

}

foreach i of numlist 51/98 {


generate double nfwgt`i' = cwgt0

}

save hints2007, replace

* Read in HINTS PR data

use "<insert file path name>\hints_puertorico.dta", clear

* Create the analysis variable

recode hc15useinternet (1=1) (2=0), generate(useinternet_r)

* Create the full sampling weight for combined data

generate double nfwgt0 = r12wgt0

* Create the replicate weights for combined data

foreach i of numlist 1/50 {


generate double nfwgt`i' = r12wgt0

}

foreach i of numlist 51/98 {


local j = `i' - 50


generate double nfwgt`i' = r12wgt`j'

}

append using hints2007

save "<insert file path name>\hints.dta", replace

erase hints2007.dta

exit, STATA clear
*Step 2: Analysis: Produce estimates for internet use prevalence
set memory 512m

use "<insert file path name>\hints.dta", clear

* Survey analysis

svyset [pw=nfwgt0], jkr(nfwgt1-nfwgt98, multiplier(0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.98 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333 0.83333)) vce(jack) mse

svy: mean useinternet_r
exit, STATA clear

