Analytic Methods for Combining HINTS Puerto Rico and HINTS 2007

This text provides instruction on the correct way to combine HINTS Puerto Rico with HINTS 2007 and also furnishes the proper SAS and SUDAAN syntax to analyze the combined data set. The method outlined here may also be used to combine other replicate weight dataset that have differing amounts of replicate weights. For more detailed information on combining datasets, see http://hints.cancer.gov/docs/HINTS_Data_Users_Handbook-2008.pdf. 

We begin with some general recommendations on combining survey datasets. All variables that will be used in the analysis should first be examined within each separate dataset, using the formats that are specific to that year. This is important because it is not uncommon for subtle changes to be enacted between survey year iterations. The following should always be checked when data is first combined:
1. Variable names, values, and formatting

2. Variable universes, skip patterns, and labels
In some cases, these will be uniform within different survey iterations and no changes will be needed. In other cases, variable recodes will be necessary to make the definitions consistent across years. A good way to verify consistency is to check descriptive statistics (frequencies, means, etc.) for the variables or recodes for drastic differences. 


Once the variables and recodes are constructed, we must modify the base and replicate weights. Unless weighted sample counts (i.e., population estimates) are desired, no changes are required to be made to the base weights. For instance, when combining two national surveys of similar sample size, a good approximation would be to divide the weights by two. We will not make any adjustments in our case, it will suffice to say that our population is the United States and Puerto Rico. For a general method on the proper reweighting of pooled surveys, see Korn and Graubard (1999).

We are now ready to define a new set of replicate weights within our pooled data set. Since HINTS 2007 has 50 replicate weights and HINTS PR contains 48 replicate weights, the resulting pooled data set should contain 98 replicate weights. For observations from HINTS 2007, the first 50 replicate weights (CWGT1-CWGT50) will be the same, while replicate weights 51-98 will be equal to the base weight (CWGT0). On the other hand, for observations from Puerto Rico, the first 50 replicate weights will be the base weight (R12WGT0), while replicate weights 51-98 will be equivalent to the 48 original replicate weights (R12WGT1-R12WGT48). The example code below uses the full HINTS 2007 dataset (mail and phone modes combined). Since the numbers of jackknife replicate weights differ, each dataset is analyzed with a different jackknife multiplier: 0.98 in HINTS 2007 and 0.83 in HINTS Puerto Rico. Therefore, additional syntax is required to properly apply the correct jackknife multiplier to each replicate weight in the combined.
/*Sample code to combine and analyze the 2007 hints mainland data with the Puerto Rico data*/

/*Step 1: Combining the two SAS files together*/
/*Different variable names for same outcomes in the two files need to be renamed to the same ones before the merge*/

/*create a survey source variable, type, in each file before the combination to distinguish the source of each observation*/

proc format;

   value typef


1='HINTS 2007'


2='HINTS Puerto Rico'


;

run;

data hints;

   set hints07 hintspr;

   /*Weight Arrays*/

   array wgts07[50]     cwgt1-cwgt50;

   array PRwgts07[48] r12wgt1-r12wgt48;  

   array newWgts[98]   nfwgt1-nfwgt98;

   attrib type label="Survey Type" format=typef.;

   /*Define new replicate and base weights*/

   if type=1 then do;***HINTS 2007;


nfwgt0=cwgt0;


do i=1 to 50;


  newWgts[i]=wgts07[i];


  end;


do j=1 to 48;


  newWgts[j+50]=cwgt0;


  end;

      end;

else if type=2 then do;***HINTS PR;

   nfwgt0=r12wgt0;

   do p=1 to 48;


newWgts[50+p]=PRwgts07[p];


end;

   do q=1 to 50;


newWgts[q]=r12wgt0;


end;

end;


label nfwgt0="Final full-sample weight";

attrib nfwgt1-nfwgt98 label="Final sample replicate weights";

drop cwgt0 r12wgt0 cwgt1-cwgt50 r12wgt1-r12wgt48;
run;

/*Step 2: Analysis: Using SUDAAN and SAS 9.2 to produce estimates for internet use prevalence by survey source and gender*/
/*Recode the internet use variable to dummy variable UseInternet_R */

data hints;

   set hints;

   if      HC15UseInternet=1 then UseInternet_R=1;

   else if HC15UseInternet=2 then UseInternet_R=0;

   else                           UseInternet_R=.;

   run;

/*SUDAAN example code*/
proc descript data=hints design=jackknife;

   weight nfwgt0;

   jackwgts nfwgt1-nfwgt98;

   jackmult 50*0.98 48*0.83;  /*the factors can also be listed separately like the SAS example below*/
   var UseInternet_R;

   class surveysource Gender;
   table surveysource Gender;
   run;

/*replace 0.83 to the correct adjjack factor for the PR data*/

/*SAS 9.2 example code*/
proc surveymeans data=hints varmethod=jackknife;

weight nfwgt0;

repweights nfwgt1-nfwgt98/jkcoefs=

;

var UseInternet_R;

run;

/*Alternative SAS 9.2 example code*/
/*Save the jackknife factors into a user defined data set (say jkcoefficients) first with the SAS default variable name jkcoefficient for the factors;*/

proc surveymeans data=hints varmethod=jackknife;

weight nfwgt0;

repweights nfwgt1-nfwgt98/jkcoefs=jkcoefficients;/* The variable name the adjustment factors must be jkcoefficient in this file*/

var UseInternet_R;

run;

